

減価償却計算プログラムの説明  
シート編

V 5 . 2

平成 2 1 年 2 月

小林誠一郎

## 減価償却のプログラムについて

減価償却方法が変わってプログラムの変更を今回何度か行いました。今、各セルにある式を見ても、計算方法が増えて、場合分けが多くなり、どう計算しているか、自分でもわかりにくくなっています。思いだせる今のうち記録を取っておくことが必要だと思いました。

平成19年度の税制改正がある前の段階までを一区切りとし、その後改正部分を加えるという、実際の作業に応じた説明とします。

データ入力画面は図1のとおりです。個人の税務申告書の様式です。提出用書類には期首の残高欄がありませんが、計算上では必ず必要ですので、一番最後に一列あけてT列に期首残高欄を設けます。計算方法指定の欄をA列、資産の種類を指定する欄をB列とします。書類印刷ではA列B列は非表示にして印刷します。

図1

計算方法	資産種類	減価償却資産等の名称等 (繰延資産を含む)	面積又は数量	取得年月	(イ)取得価額	(ロ)償却の基礎になる金額	償却方法	耐用年数	(ハ)償却率又は定率	(ニ)本年中の償却期間	(ホ)本年分の普通償却費 ロ×ハ×ニ
A	B	C	D	E	F	G	H	I	J	K	L

  

(ホ)本年分の普通償却費 ロ×ハ×ニ	(ヘ)割増(特別)償却費	(ト)本年分の償却費合計 ホ+ヘ	(チ)平等率 用割合	(リ)本年分の必要経費算入額 ト×チ	(ス)未償却残高 (期末残高)	摘要	*	未償却残高 (期首残高)
L	M	N	O	P	Q	R	S	T

図の中のアルファベットは列名を表わします。データ入力の1番上の行は7行目になり、106行目までの100件を使用。黄色いセルには手入力が必要な部分、白い部分は計算式で表示できる部分です。

データ入力画面と実際に計算する画面は別にしてあります。

図2 シート「計算表」



A	計算方法	
B	資産種類	
C	名称等(含繰延)	
D	数量	
E	年月	
F	取得価額	
G	基礎になる金額	=AA7
H	方法	=IF(A7="", "", IF(A7=2, "定率", "定額"))
I	年数	
J	償却率	=IF(OR(A7="", I7=""), "", IF(A7<>2, VLOOKUP(I7, 耐用年数, 2), VLOOKUP(I7, 耐用年数, 3)))
K	月数	
L	償却費	=AE7
M	割増償却費	
N	償却費合計	=IF(L7="", "", L7+M7)
O	割合	
P	算入額	=IF(N7="", "", ROUNDDOWN(N7*O7/100, 0))
Q	未償却残高	=IF(P7="", "", T7-N7)
R	摘要	
S	*	
T	未償却残高	
U		
V	償却の基礎になる金額のもと	=IF(\$A7="", "", VLOOKUP(\$A7, 定数表, 2))
W	残存割合 a	=IF(\$A7="", "", VLOOKUP(\$A7, 定数表, 3))
X	残存割合 b	=IF(\$A7="", "", VLOOKUP(\$A7, 定数表, 4))
Y		
Z	牛馬の特例	=IF(F7="", "", IF(F7*X7>=100000, 100000, F7*X7))
A	償却の基礎になる金額	=IF(F7="", "", IF(A7=2, T7, IF(Z7=100000, F7-100000, ROUNDDOWN(F7*V7, 0))))
B	単純償却費	=IF(OR(F7="", I7=""), "", ROUNDDOWN(AA7*J7*K7/12, 0))
C	単純今期末	=IF(OR(F7="", I7=""), "", T7-AB7)
D	償却限度額	=IF(F7="", "", IF(Z7=100000, Z7, ROUNDDOWN(F7*W7, 0)))
E	当期償却額	=IF(AC7>=AD7, AB7, IF(T7>AD7, T7-AD7, 0))

前のページの式の中に、VLOOKUPの中に耐用年数と定数表という名称が出てきますが、これは「参照表」というシートに耐用年数と定数表という範囲名で設定しています。

耐用年数 \$A\$3:\$C\$102 1年から100年分で定額法と定率法の方です。

定数表 \$E\$3:\$H\$12 定額法が9種類、定率用が1種類です。

図5

	A	B	C	D	E	F	G	H	I	J	K	L
1		定額法	定率法		計算方法							
2	耐用年数	償却率	償却率		定数表	基礎の分	償却限度	特例分				
3	1	1	0.2		1	0.9	0.05	0				
4	2	0.5	0.684		2	0	0.05	0				
5	3	0.333	0.536		3	0.6	0.4	0.4				
6	4	0.25	0.438		4	0.8	0.2	0.2		耐用年数	\$A\$3:\$C\$102	
7	5	0.2	0.369		5	0.9	0.1	0.1		定数表	\$E\$3:\$H\$12	
8	6	0.166	0.319		6	0.5	0.5	0.5				
9	7	0.142	0.28		7	0.7	0.3	0.3				
10	8	0.125	0.25		8	0.7	0.3	0				
11	9	0.111	0.226		9	0.95	0.05	0				
12	10	0.1	0.206		10	1	0	0				
13	11	0.09	0.189									
14	12	0.083	0.175		1 定額法	一般						
15	13	0.076	0.162		2 定率法							
16	14	0.071	0.152		3 定額法	牛	小運搬使役用					
17	15	0.066	0.142		4 定額法	牛	繁殖用乳用牛、種付用の役肉用牛					
18	16	0.062	0.134		5 定額法	牛	種付用の乳用牛					
19	17	0.058	0.127		6 定額法	牛	農業使役用その他用					
20	18	0.055	0.12		4 定額法	馬	小運搬使役用、繁殖用、競争用					
21	19	0.052	0.114		5 定額法	馬	種付用					
22	20	0.05	0.109		7 定額法	馬	農業使役用その他用					
23	21	0.048	0.104		8 定額法	豚						
24	22	0.046	0.099		9 定額法	綿羊、やぎ、果樹その他の植物						
25	23	0.044	0.095		10 定額法	営業権、繰延資産						
26	24	0.042	0.092									
27	25	0.04	0.088									

#### 範囲名についての余談。(給与計算プログラムから一部抜粋)

プログラムの修正に関連しての不具合です。賞与遡及処理の保険計算部分の転記作業を座標指定でしていたため2行挿入した結果、不具合発生。マクロの Sub 転記賞与遡及税額2()の中の Range("D67:K69").Select を Range("D69:K71").Select に変更します。

作業を始めるときには、範囲名をつけて作業をすることが多いです。行列の挿入削除があってもマクロの中には影響を受けないからです。ある程度固まってきたら範囲名をつけずにマクロを書くことが多くなります。これには2つの理由があります。範囲名をつけることが煩わしいのが一つ。範囲名をつけると場所がすぐわからないのが一つ。マクロを読む時に範囲名であればその確認がすぐ出来ないの、内容がわかりにくいからです。範囲名を使わずにセル名でマクロを書いてあれば作業内容がわかりやすいと思うのです。

でも、上記のような不具合が発生するのはセル指定のためです。同じような作業を4枚の表でしていますが、問題の表以外は範囲名で作業をしていたので問題は起こらないのです。範囲名を活用するのがやはり正しい方法でしょうか。表計算ソフトはどこでなにをしているのかが分かりにくいです。セルにある式の解析が簡単ではないからです。マクロの部分は流れを追うのは簡単で、命令の意味が分かりさえすれば内容も理解しやすいです。

19年度の改正のときの修正ではこの耐用年数と定数表という範囲名を変更するだけで、計算表の式の方は何も変更する必要はありませんでした。

耐用年数 \$A\$3:\$G\$102 4列分のデータが増えました。

定数表 \$I\$3:\$L\$23 新旧合わせて21種類になりました。

範囲名は有効な方法ということになりますね。

計算方法は10種類あります。

			残存割合	償却可能限度
1	定額法	一般	10%	5%
2	定率法			5%
3	定額法	牛 小運搬使役用	40%	40%
4	定額法	牛 繁殖用乳用牛、種付用の役肉用牛	20%	20%
5	定額法	牛 種付用の乳用牛	10%	10%
6	定額法	牛 農業使役用その他用	50%	50%
4	定額法	馬 小運搬使役用、繁殖用、競争用	20%	20%
5	定額法	馬 種付用	10%	10%
7	定額法	馬 農業使役用その他用	30%	30%
8	定額法	豚	30%	30%
9	定額法	綿羊、やぎ、果樹その他の植物	5%	5%
10	定額法	営業権、繰延資産	0%	0%

牛馬については 残存価額が10万円を超える時は10万円

この表を償却の基礎になる金額を算出するための定数表を次のように準備します。

定数表 定額法が9種類、定率用が1種類です。

図6

計算方法	基礎の分	償却限度	特例分
1	0.9	0.05	0
2	0	0.05	0
3	0.6	0.4	0.4
4	0.8	0.2	0.2
5	0.9	0.1	0.1
6	0.5	0.5	0.5
7	0.7	0.3	0.3
8	0.7	0.3	0
9	0.95	0.05	0
10	1	0	0

V列に W列に X列に

図4の計算式の中の最初にあるIF関数は、そこにあるべきデータがないときには、何も表示しないというもの。見た目をきれいにしたりするときに良く使います。

**G列の償却の基礎になる金額はA A列で算出されます。**

その過程は次のようになります。

U列からA A列まで再掲します。

図 7

U		
V	償却の基礎になる 金額のもと	=IF(\$A7="", "", VLOOKUP(\$A7,定数表,2))
W	残存割合 a	=IF(\$A7="", "", VLOOKUP(\$A7,定数表,3))
X	残存割合 b	=IF(\$A7="", "", VLOOKUP(\$A7,定数表,4))
Y		
Z	牛馬の特例	=IF(F7="", "", IF(F7*X7>=100000,100000,F7*X7))
A A	償却の基礎になる 金額	=IF(F7="", "", IF(A7=2,T7,IF(Z7=100000,F7-100000,ROUND(DOWN(F7*V7,0))))))

U列、Y列は空白です。W列、X列、Y列には順番にA列で指定した計算法に従い、順番に定数表の2列目、3列目、4列目の数字が入ります。

A A列の式ですが、

$$=IF(F7="", "", IF(A7=2, T7, IF(Z7=100000, F7-100000, ROUNDDOWN(F7*V7, 0))))$$

F列（取得価額）が空白の場合は何も表示しない。金額の記入があった場合は、次のIF関数で、A列が2の場合（定率法ということです。）は、T列を表示します。T列には期首残高を入力する設定になっています。A列が2以外の時は、次のIF関数が働き、Z列が100000であった場合、取得価額から10万円を差し引いた金額が償却の基礎になる金額になります。Z列が100000でない場合はF列（取得価額）にV列（この列には100%から残存割合を差し引いた数値が設定されています。）を掛け合わせたものが、償却の基礎になる金額になります。

前をページに戻るのも苦勞なので再掲です。

定額法

$$\text{償却限度額} = \text{償却の基礎になる金額} * \text{償却率}$$

$$\text{償却の基礎になる金額} = \text{取得価額} - \text{残存価額}$$

$$\text{残存価額} = \text{取得価額} * \text{残存割合}$$

ただし、牛や馬については残存価額が10万円を超える場合は10万円とする。

定率法

$$\text{償却限度額} = \text{償却の基礎になる金額} * \text{償却率}$$

$$\text{償却の基礎になる金額} = \text{前期末未償却残高}$$

Z列にはF列（取得価額）とX列（特例の残存割合）を掛け合わせたもの入るわけですが、牛馬以外の残存割合を定数表で 0 にしてあることで、牛馬の場合について残存価額が10万円を超える場合は10万円とするということが実現できます。

償却の基礎になる金額 G列 が終了しました。

H列は 償却方法を文字で記載する欄 =IF(A7="", "", IF(A7=2, "定率", "定額"))

I列は 資産に応じた耐用年数を入力するところ。

J列は 耐用年数に応じた償却率

=IF(OR(A7="", I7=""), "", IF(A7<>2, VLOOKUP(I7, 耐用年数, 2), VLOOKUP(I7, 耐用年数, 3)))

A列が2以外の時は定額法の数値を表示し、それ以外(2の時)は定率法の数値を表示

A列とかI列かどちらかが空白の時は、何も表示しない。計算は不能になります。

K列は 当年の利用月数を入力するところ。

L列は今年の償却費が入る欄です。A E列で計算されます。

図 8

A	償却の基礎になる金額	= IF(F7="", "", IF(A7=2, T7, IF(Z7=100000, F7-100000, ROUNDDOWN(F7*V7, 0))))
B	単純償却費	=IF(OR(F7="", I7=""), "", ROUNDDOWN(AA7*J7*K7/12, 0))
C	単純今期末	=IF(OR(F7="", I7=""), "", T7-AB7)
D	償却限度額	=IF(F7="", "", IF(Z7=100000, Z7, ROUNDDOWN(F7*W7, 0)))
E	当期償却額	=IF(AC7>=AD7, AB7, IF(T7>AD7, T7-AD7, 0))

A B列 単純償却費 = 償却の基礎になる金額 \* 償却率 \* (利用月数/12)  
端数は切り捨て処理

A C列 単純今期末 = 期首未償却残高 - 単純償却費

A D列 償却限度額 = 取得価額 \* 残存割合

但し、牛馬の場合 10万円を超える場合は10万円となります。

A E列の式ですが、

=IF(AC7>=AD7, AB7, IF(T7>AD7, T7-AD7, 0))

A C列(単純今期末)がA D列(償却限度額)以上であればA B列(単純償却費)が今期の償却額になります。そうでない場合でも、T列(期首未償却残高)がA D列(償却限度額)より大きければ、T列(期首未償却残高)からA D列(償却限度額)を差し引いたものが今期の償却額になります。これら以外の場合は今期の償却額は 0 になります。

本年分の普通償却費の金額 L列 が終了しました。

M列は割増償却費を入力するところ。

N列はL列とM列を加算したもの。当期の償却費

正確な割増償却には対応していません。加算計算と残高計算はしますが未償却残高のチェックはしていません。



O列は事業専用割合を入力するところ。

P列はN列とM列を乗算し、本年分の必要経費を算出する欄。

Q列が今年の未償却残高になります。

T列（期首未償却残高）からN列（当期の償却費）を減算しています。

これで初めの減価償却計算表の説明が終了しました。

### 次が改正分のものです。

平成19年度の税制改正により減価償却の計算方法が変わりました。

旧定額法と旧定率法で償却限度額に達した資産は翌事業年度以後は、次の計算式で求められる金額を償却限度額として、残存簿価1円まで償却します。

$$\text{償却限度額} = (\text{取得価額} - \text{取得価額の95\%相当額} - 1\text{円}) * \text{各事業年度の月数}/60$$

この計算ファイルでは 各事業年度の月数/60 を 1/5 で計算しています。

定額法（平成19年4月1日以後）

次の計算式で求められる金額を償却限度額とし、残存価額が1円になるまで償却を行なう。

$$\text{償却限度額} = \text{取得価額} * \text{定額法の償却率}$$

定率法（平成19年4月1日以後）

1. まず、次の2つの式で調整前償却額と償却保証額の金額を求める。

$$\text{調整前償却額} = \text{期首未償却残高} * \text{定率法の償却率}$$

$$\text{償却保証額} = \text{取得価額} * \text{耐用年数に応じた保証率}$$

2. 調整前償却額と償却保証額の金額を比較し、当期の償却限度額を求める。

(1) 調整前償却額 ≧ 償却保証額の場合

$$\text{償却限度額} = \text{調整前償却額}$$

(2) 調整前償却額 < 償却保証額の場合

$$\text{償却限度額} = \text{改定取得価額} * \text{改定償却率}$$

ここで改定取得価額には変更時点での期首未償却残高を用い、残存価額1円になるまで償却を行なう。

牛馬の場合の償却限度額に達したものは

$$\text{償却限度額} = (\text{残存価額} - 1\text{円}) * 1/5$$

計算方法が1から10まででしたが、

\* 新定額法を11、新定率法を12、限度後償却法（均等償却）を13、改定定額法（改定償却）を14として追加。（V30）

\* 償却の基礎になる金額の記載方法の変更（V40）

\* 定率法の償却保証額を記載するために2段書の印刷部分追加（V50）

\* 牛馬の場合の償却限度額に達したものの計算方法を追加（V52）

15から21までを追加した。一覧表が図9です。

名称は限度後定額法は均等償却法、改定定額法は改定償却法と呼ぶのが正しいかな。

図 9

1	旧定額法	一般	
2	旧定率法	一般	
3	旧定額法	牛	小運搬使役用
4	旧定額法	牛	繁殖用乳用牛、種付用の役肉用牛
5	旧定額法	牛	種付用の乳用牛
6	旧定額法	牛	農業使役用その他用
4	旧定額法	馬	小運搬使役用、繁殖用、競争用
5	旧定額法	馬	種付用
7	旧定額法	馬	農業使役用その他用
8	旧定額法	豚	
9	旧定額法	綿羊、やぎ、果樹その他の植物	
10	旧定額法	営業権、繰延資産、即時償却、一括償却	
11	新定額法	一般	
12	新定率法	一般	
13	限度後定額法		
14	改定定額法		
15	限度後定額法	牛 3	
16	限度後定額法	牛 4	馬 4
17	限度後定額法	牛 5	馬 5
18	限度後定額法	牛 6	
19	限度後定額法		馬 7
20	限度後定額法		豚 8
21	限度後定額法		綿羊 9

牛馬のことをもっと先にしっかりと検討すればもうすこし簡単にできたのかな。

14があいだにはいらないと図10の計算式も簡単になりましたね。遅すぎる。( ; \_ \_ )

図10が減価償却V52の「計算表」シートの計算式です。

列名に色がついていますが、黄はV40、V52で変更した部分。赤はV30、V52で変更した部分。緑はV30で変更した部分。

式の中の変更部分は赤にしています。

限度後定額法のためAG列、AH列、AI列

定率法の償却保証額のためAK列、AL列

図 1 0

A	計算方法	
B	資産種類	
C	名称等	
D	数量	
E	年月	
F	取得価額	
G	償却の基礎 になる金額	=IF(OR(A7=13,A7>=15),AD7,AA7)
H	方法	=IF(A7=""," ",IF(A7=2,"旧定率",IF(A7=11,"定額",IF(A7=12,"定率",IF(OR(A7=10,A7=13,A7>=15),"-",IF(A7=14,"改定額","旧定額"))))))
I	年数	
J	償却率	=IF(OR(A7=" ",I7="")," ",IF(A7=2,VLOOKUP(I7,耐用年数,3),IF(A7=12,VLOOKUP(I7,耐用年数,5),IF(A7=11,VLOOKUP(I7,耐用年数,4),IF(A7=14,VLOOKUP(I7,耐用年数,6),VLOOKUP(I7,耐用年数,2))))))
K	月数	
L	償却費	=AE7
M	償却費	
N	償却費合計	=IF(L7=""," ",L7+M7)
O	割合	
P	算入額	=IF(N7=""," ",ROUNDDOWN(N7*O7/100,0))
Q	未償却残高	=IF(F7<>"",T7-N7,T7)
R	摘要	=IF(S7="*", "14に変更",IF(AND(A7=13,P7<>0),"均等償却",IF(AND(A7=14,P7<>0),"改定償却",IF(AND(A7>=15,P7<>0),"均等償却",""))))
S	*	=IF(AL7=1,"*", "")
T	未償却残高	
U		
V	償却の基礎 になる金額 のもと	=IF(\$A7=""," ",VLOOKUP(\$A7,定数表,2))
W	残存割合 a	=IF(\$A7=""," ",VLOOKUP(\$A7,定数表,3))
X	残存割合 b	=IF(\$A7=""," ",VLOOKUP(\$A7,定数表,4))
Y		
Z	牛馬の特例	=IF(F7=""," ",IF(F7*X7>=100000,100000,F7*X7))
A	償却の基礎 になる金額	=IF(F7=""," ",IF(OR(A7=2,A7=12),T7,IF(A7=14,U7,IF(Z7=100000,F7-100000,ROUNDDOWN(F7*V7,0))))))

A		
B	単純償却費	=IF(OR(F7="",I7=""),"",ROUND(DOWN(AA7*J7*K7/12,0)))
A		
C	単純今期末	=IF(OR(F7="",I7=""),"",T7-AB7)
A		
D	償却限度額	=IF(F7="", "", IF(Z7=100000, Z7, IF(OR(A7=11, A7=12, A7=14), 1, ROUNDUP(F7*W7, 0))))
A		
E	当期償却額	=IF(OR(A7=13, A7>=15), A17, IF(AC7>=AD7, AB7, IF(T7>AD7, T7-AD7, 0)))
A		
F		
A	限度後	
G	償却	=IF(OR(A7=13, A7>=15), ROUNDUP((AD7-1)/5, 0), 0)
A		
H		=IF(AND(AG7<>0, T7-AG7<=0), 1, 0)
AI	当期償却額	=IF(AH7=1, T7-1, AG7)
A		
J		
A	償却保証	
K	額(取得価額 *保証率)	=IF(OR(A7=12, A7=14), ROUNDUP(F7*VLOOKUP(I7, 耐用年数, 7), 0), 0)
A	下回る時	
L	は1	=IF(AND(A7=12, AE7<AK7), 1, 0)

A D列 V1.6で切り上げに変更 AG列 V4.0で切り上げに変更

セルA L 6の式 =IF(ISERROR(SUM(AL7:AL106)), "1000", SUM(AL7:AL106))

図11 参照表の範囲名を拡大、移動

	A	B	C	D	E	F	G	H	I	J	K	L
1		旧定額法	旧定率法	定額法	定率法	定率法	定率法		計算方法			
2	耐用年数	償却率	償却率	償却率	償却率	改定償却率	保証率		定数表	基礎の分	償却限度	特例分
3	1	0.2							1	0.9	0.05	0
4	2	0.5	0.684	0.5	1	0	0		2	0	0.05	0
5	3	0.333	0.536	0.334	0.833	1	0.02789		3	0.6	0.4	0.4
6	4	0.25	0.438	0.25	0.625	1	0.05274		4	0.8	0.2	0.2
7	5	0.2	0.369	0.2	0.5	1	0.06249		5	0.9	0.1	0.1
8	6	0.166	0.319	0.167	0.417	0.5	0.05776		6	0.5	0.5	0.5
9	7	0.142	0.28	0.143	0.357	0.5	0.05496		7	0.7	0.3	0.3
10	8	0.125	0.25	0.125	0.313	0.334	0.05111		8	0.7	0.3	0
11	9	0.111	0.226	0.112	0.278	0.334	0.04731		9	0.95	0.05	0
12	10	0.1	0.206	0.1	0.25	0.334	0.04448		10	1	0	0
13	11	0.09	0.189	0.091	0.227	0.25	0.04123		11	1	0	0
14	12	0.083	0.175	0.084	0.208	0.25	0.0387		12	1	0	0
15	13	0.076	0.162	0.077	0.192	0.2	0.03633		13		0.05	0
16	14	0.071	0.152	0.072	0.179	0.2	0.03389		14	1	0	0
17	15	0.066	0.142	0.067	0.167	0.2	0.03217		15		0.4	0.4
18	16	0.062	0.134	0.063	0.156	0.167	0.03063		16		0.2	0.2
19	17	0.058	0.127	0.059	0.147	0.167	0.02905		17		0.1	0.1
20	18	0.055	0.12	0.056	0.139	0.143	0.02757		18		0.5	0.5
21	19	0.052	0.114	0.053	0.132	0.143	0.02616		19		0.3	0.3
22	20	0.05	0.109	0.05	0.125	0.143	0.02517		20		0.3	0
23	21	0.048	0.104	0.048	0.119	0.125	0.02408		21		0.05	0
24	22	0.046	0.099	0.046	0.114	0.125	0.02296					

耐用年数 \$A\$3 : \$G\$102 定数表 \$I\$3 : \$L\$23 図11

図 1 2

	AC	AD	AE	AF	AG	AH	AI	AJ	AK	AL	AM	AN
1					19年8月追加分				新規定率法			
2		V1.0で切り上げに変更			19年9月31日まで分				償却保証額(取得価額*保証率)			
3	単純今期末	償却限度額	当期償却額		限度後償却		当期償却額		下回る時は1			
4		V5.0で修正			切り上げ				V4.0で切り上げに家業の60で14%修正)			
6				V51 動植物でG列H列R列AE列AG列を変更							V52で式変更	
7					0	0	0		0	0		
8					0	0	0		0	0		
9					0	0	0		0	0		
10					0	0	0		0	0		

図 1 2 の A G 列以降が新しい方法に対応するために増やした計算式ですが、その説明。

図 1 3

	限度後 償却	
AG		=IF(OR(A7=13,A7>=15),ROUNDUP((AD7-1)/5,0),0)
AH		=IF(AND(AG7<>0,T7-AG7<=0),1,0)
AI	当期償却額	=IF(AH7=1,T7-1,AG7)

均等償却法を償却限度額に達した資産の残存価額を5年間で償却します。というように解釈してV30を作成しました。(あとで、よく書類をみると、使用月数/60 だったので、微妙なところで食い違いが起こるかもしれない。( ; \_ \_ )

A G 列 A 列が 1 3 かもしくは 1 5 以上の場合は、A D 列から 1 を減算したものを更に 5 で除算したものとなり(切り上げ処理、小数点で)、そうでない場合は 0 となります。

(均等償却法の場合は数値があり、そうでない場合は 0 となる。)

A H 列 A G 列が 0 ではなく、かつ、T 列から A G 列を減算した結果が 0 以下の場合は 1 となり、そうでない場合は 0 となります。

A I 列 A H 列が 1 ならば、T 列から 1 を減算したものを、そうでない場合は A G 列になります。

具体的な内容は

A G 列 均等償却法の1年間の償却限度額を計算しています。

(償却限度額 - 1円) \* 1 / 5 で1年間の償却限度額を現します。

均等償却法でない場合は 0 となります。

A H 列 均等償却の場合で、期首未償却残高から今期の償却限度額を差し引けないものをチェック、1を表示。

A I 列 A H 列が 1 ならば、今年最後の償却なので、残高の1円を減算したものが、今年の償却額であり、そうでない場合は償却限度額を差し引けることになる。均等償却法の場合のみ償却限度額が表示されることとなります。

図 1 4

AJ		
AK	償却保証額(取得価額 * 保証率)	=IF(OR(A7=12,A7=14),ROUNDUP(F7*VLOOKUP(I7,耐用年数,7),0),0)
AL	下回る時は 1	=IF(AND(A7=12,AE7<AK7),1,0)

A K 列と A L 列は、定率法の償却保証額を計算チェックする欄です。

A K 列

計算方法が 1 2 と 1 4 の場合は、F 列(取得価額) \* 保証率(小数点以下、切上処理) それ以外のときは 0 になります。

A L 列

定率法で計算した場合で、A E 列(今期の償却額)が A K 列(償却保証額)を下回ったときに、1 を表示します。改定償却法でなければいけない合図です。

**最初からもう一度チェックです。**

G 列

図 1 5

G	償却の基礎になる金額	=IF(OR(A7=13,A7>=15),AD7,AA7)
---	------------	-------------------------------

均等償却法の場合には、償却の基礎になる金額は昔の償却限度額になります。

A	償却限度額	=IF(F7="", "", IF(Z7=100000,Z7,IF(OR(A7=11,A7=12,A7=14),1,ROUNDUP(F7*W7,0))))
---	-------	-------------------------------------------------------------------------------

新しい計算方法は償却限度額が 1 円になるので、その部分を加えました。

A	償却の基礎になる金額	=IF(F7="", "", IF(OR(A7=2,A7=12),T7,IF(A7=14,U7,IF(Z7=100000,F7-100000,ROUNDDOWN(F7*V7,0))))))
---	------------	------------------------------------------------------------------------------------------------

旧定率法と新定率法は T 列(期首未償却残高)が償却の基礎になる金額です。

改定償却法の場合は U 列(V 5 0 以降設定、改定取得価額)。ここにはこの変更時の期首未償却残高を入力し、翌年以降も引き継ぎます。

H 列、I 列、J 列

H	方法	=IF(A7="", "", IF(A7=2, "旧定率", IF(A7=11, "定額", IF(A7=12, "定率", IF(OR(A7=10,A7=13,A7>=15), "- ", IF(A7=14, "改定額", "旧定額"))))))
I	年数	
J	償却率	=IF(OR(A7="", I7=""), "", IF(A7=2, VLOOKUP(I7,耐用年数,3), IF(A7=12, VLOOKUP(I7,耐用年数,5), IF(A7=11, VLOOKUP(I7,耐用年数,4), IF(A7=14, VLOOKUP(I7,耐用年数,6), VLOOKUP(I7,耐用年数,2))))))

均等償却法の場合は、償却方法に - を表示させるのが目的です。

償却率は 4 列増加。 耐用年数 \$ A \$ 3 : \$ G \$ 1 0 2。

A	B	C	D	E	F	G
	旧定額法	旧定率法	定額法	定率法	定率法	定率法
耐用年数	償却率	償却率	償却率	償却率	改定償却率	保証率

K列、L列、M列、N列、O列、P列

その中の式はこれまでと同じで変わりません。が、L列から呼び出されるA E列は

A		=IF(OR(A7=13,A7>=15),AI7,IF(AC7>=AD7,AB7,IF(T7>AD7,T7-AD7,0)))
E	当期償却額	

従来のものに、最優先として、均等償却法の場合はA I列を表示するように変更しています。その後の優先順位は同じです。

Q列、R列、S列、T列、U列

Q	未償却残高	=IF(F7<>"",T7-N7,T7)
R	摘要	=IF(S7="*", "14に変更",IF(AND(A7=13,P7<>0),"均等償却",IF(AND(A7=14,P7<>0),"改定償却",IF(AND(A7>=15,P7<>0),"均等償却",""))))
S	*	=IF(AL7=1,"*", "")
T	未償却残高	
U		

Q列

以前の未償却残高 =IF(P7="", "", T7-N7) これはP列（経費算入額）がないときは何も表示しない設定です。ある時には、差し引きした結果の残高が表示されます。今回の =IF(F7<>"", T7-N7, T7) はP列は関係なく、F列が空白でない場合、上と同じ結果になります。F列が空白の場合はT列を表示。この意味がわからない。変更した時期も思い出せない。特に間違いはないでしょう。？。

R列（適用欄）にはいろいろな文字を表示させます。

S列 A L列が1のとき \* 印を表示。同時にQ列は「14に変更」と表示。  
変更すべきデータを目で確認できるようにしたものです。

T列 以前と同じ。

U列 定率法を改定償却に変えたときの再取得価額を入力する欄です。

A L列

定率法の場合で、A E列（今期の償却額）がA K列（償却保証額）を下回ったときに、1を表示します。

切り替えるべきに注意するような仕組みです。マクロのほうでも

セルA L 6の式 =IF(ISERROR(SUM(AL7:AL106)), "1000", SUM(AL7:AL106))

をチェックしてメッセージを出すようにしています。

## その他のシートの説明です。

このプログラムにはシートが全部で34枚あります。

"MENU" "a" から "z" "説明表" "計算表" "印刷表" "印刷表 2" "data" "転記" "参照表"

スタートは"MENU"シート。図16

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
1														
2				○減価償却費の計算		V52	全部計算	全部印刷	全部年度更新転記					
3				事業所名			事業所名		連続計算範囲指定					
4				1 a		14 n			<input type="checkbox"/> から <input type="checkbox"/> まで 一部計算 1 1					
5				2 b		15 o			連続印刷範囲指定 <input type="checkbox"/> から <input type="checkbox"/> まで 一部印刷 1 1 範囲指定は数値でします。					
6				3 c		16 p								
7				4 d		17 q								
8				5 e		18 r								
9				6 f		19 s								
10				7 g		20 t								
11				8 h		21 u								
12				9 i		22 v								
13				10 j		23 w								
14				11 k		24 x								
15				12 l		25 y								
16				13 m		26 z								
17									各事業所へ					
18									K5000					
19				住所入力欄を下のほうに										
20				設けました。										
21														
22														
23														
24														
25														
26														
27				事業所名		住所 (E列に入力します)			納税者番号 (J列に入力します)					
28				1 a		山川			123456					
29				2 b										
30				3 c										

このシートの仕掛けはセル I 7 とセル K 7。 一部計算の数値取得に利用。

セル I 11 とセル K 11。 一部印刷の数値取得に利用。

これらの数値はマクロで利用します。

"a" から "z" のシートは全部で26枚。

各事業所データが入るシート。同じものです。シートの名称が違うだけ。"MENU"に入力された事業名や住所、納税者番号が表示されるセルが4か所あります。ほかに次の式。

セル X 1 の式 =CELL("FILENAME",C1)

セル X 2 の式 =MID(CELL("FILENAME",C1),SEARCH("]",CELL("FILENAME",C1),1)+1,100)

この2つの関数でセル X 2 にそのシートの名前が表示されます。入力画面と計算画面とのデータの転記に利用します。また、転記するべきデータの数値のカウントをセル J 2 =COUNT(F:F) で行っています。F列は取得価額でデータは空白なく入力することの意味がここにあります。ここにある数だけ7行目から数えて転記します。空白があればデータの抜けが起こります。

セル P 2 =COUNT(P:P)-1 印刷時のチェックに利用。データが不完全な状態で印刷すると印刷表が壊れたことがあったので、その防止の意味でチェック欄を設定しました。マクロで利用。

"説明表"シート

特に説明はないですね。ただのシートです。



## "計算表"

これは詳しく説明しましたね。

## "印刷表"

入力画面にあるデータを提出用に印刷するときに利用します。表題を作っています。

データのある画面を印刷するように設定した場合に、思わぬことでエラーが起こり、データが破壊されることがあります。とりあえず、データを転記したあとで、そこでいろいろ作業を行う。作業後のあとは、元に戻すことなく、利用データを消してしまえば元のデータはなんら影響を受けることはありません。

## "印刷表 2"

### "data"

定額法の場合、償却保証額を記載する欄が増えました。すべてのデータを"data"シートに転記すると、"印刷表 2"にそのデータを表示する式を、「印刷表 2」に埋め込む方法を取りました。小計や合計はできません。計算式だけで、小計、合計を出すのは複雑すぎるので、あきらめました。

## "参照表"

計算表と一緒に説明済でいいですね。

## "転記"

データ転記ボタン 1.3\*以前から

説明書の下の方に、データ転記ボタンを設置していますが、古いファイルからデータを取り込む時に、このシートを一時的に利用したもの。今は使っていないことになります。

バージョン1.3 はどんな形だったか覚えていません。今回、この説明を書くためにチェックしたところ、ファイルにあるシートの名称をここに取り込んでいました。

なんらかの利用価値はあると思うので、残しておこう。( ^ - ^ )

以上、シート編の終了です。

平成21年2月25日

小林誠一郎